
*** Stata Do-file for PMI Access:Use report using MIS or DHS datasets
* (see www.vector-works.org/resources/llin-use-and-access-for-pmi-countries/)
** Authors: Albert Kilian and Hannah Koenker

** Goals: calculate three indicators – 
*	1) household ownership of any ITN; 
*	2) population access to ITN; 
*	3) population use of ITN the previous night

* Step 1 - prep universal coverage indicators

* in the household data file (HR) – generate a new hhid variable to assist with merging. 
* Note: very occasionally hv001 must be multiplied by 10,000 to accomodate longer hv002 strings.

	rename hhid hhidx
	gen hhid=(hv001*1000)+hv002
	gen persid=hhid*100+hv003

	foreach var of varlist hml10_*{
	gen itn_`var' = `var'==1
	}
	egen nitn= rsum (itn*)
	
	
	label var nitn "# of ITN in household"

	gen anyitn=nitn>0
	label var anyitn "hh has any ITN"

** calculate variable for households with at least 1 ITN for 2 people (p2itn) before doing any merging, if you want	
* create de jure members 'person' from hv012
clonevar person=hv012

gen itnpers=nitn/person              
label var itnpers "ITN per person in household"

* SAVE the household data file (HR)
	save, replace

* create a small dataset with only hhid, nitn, anyitn to merge into the household member dataset
	keep hhid nitn anyitn 

	save nitn, replace

* Step 2 - OPEN the household member data set (PR). Population Access MUST be calculated in the 
* hh member dataset to do the weighting correctly for individuals (and not by hh)

* generate a separate, comparable hhid to match the datasets on
	rename hhid hhidx
	gen hhid=(hv001*1000)+hv002
	gen persid=hhid*100+hv003

* merge the PR file and the new 'nitn' file to bring the nitn variable into the household member (PR) file
	merge m:1 hhid using nitn
	drop _merge

* generate the 'stayed in the household last night' (defacto members) variable
	gen stay=hv103==1
	label var stay "stayed in house last night"
* rename the stratification variables - 
	** region (region or province); 
	** urban (urban or rural residence); 
	** ses (wealth quintile), 
	** irs (household was sprayed in the last 12 months)
	clonevar region=hv024
	clonevar urban=hv025
	clonevar ses=hv270
	clonevar irs=hv253
	clonevar month=hv006
	clonevar year=hv007
	clonevar age=hv105
* Step 3 - Still in the household member (PR) file, calculate population access using the MERG method 
	gen potuse=nitn*2
	label var potuse "potential ITN users in hh"

	egen defacto=sum(stay), by(hhid)
	label var defacto "de facto population"

	bysort hhid: gen access2 = potuse/defacto
	replace access2=1 if potuse/defacto >1
	label var access2 "proportion in hh with access"
	* gen the useitn variable 
	gen useitn=hml12==1 
** set survey weights using cluster (hv001), strata (hv024) and household sampling weight (hv005)
	svyset hv001 [pw=hv005], strata(hv024)

	** calculate national population access
	svy: mean access2 if stay==1
	** calculate stratified population access, use 'logistic' command to assess significant differences from 1st result
	svy: mean access2 if stay==1, over(region)
	svy: logistic access2 i.region if stay==1
	svy: mean access2 if stay==1, over(ses)
	svy: logistic access2 i.ses if stay==1
	svy: mean access2 if stay==1, over(urban)
	svy: logistic access2 urban if stay==1
	svy: mean access2 if stay==1, over(irs)
	svy: logistic access2 i.irs if stay==1

* Step 4 - still in the household member (PR) file, calculate the mean net use 

	
	** calculate national population use of ITNs
	svy: mean useitn if stay==1
	** calculate stratified population use
	svy: mean useitn if stay==1, over(region)
	svy: logistic useitn i.region if stay==1
	svy: mean useitn if stay==1, over(ses)
	svy: logistic useitn i.ses if stay==1
	svy: mean useitn if stay==1, over(urban)
	svy: logistic useitn urban if stay==1
	svy: mean useitn if stay==1, over(irs)
	svy: logistic useitn i.irs if stay==1
	* SAVE the household member file (PR)
	save, replace


* Step 5 - CLOSE the household member file and OPEN the household file. 

	* rename the needed stratification variables - 
	** region (region or province); 
	** urban (urban or rural residence); 
	** ses (wealth quintile), 
	** irs (household was sprayed in the last 12 months)
	clonevar region=hv024
	clonevar urban=hv025
	clonevar ses=hv270
	clonevar irs=hv253
	clonevar month=hv006
	clonevar year=hv007
	save, replace

	** calculate ownership of at least one ITN in household file
	
** set survey weights using cluster (hv001), strata (hv024) and household sampling weight (hv005)
	svyset hv001 [pw=hv005], strata(hv024)
	
	** calculate national household ownership of at least 1 ITN
	svy: mean anyitn
	** calculate stratified household ownership of at least 1 ITN
	svy: mean anyitn, over(region)
	svy: logistic anyitn i.region
	svy: mean anyitn, over(ses)
	svy: logistic anyitn i.ses
	svy: mean anyitn, over(urban)
	svy: logistic anyitn urban
	svy: mean anyitn, over(irs)
	svy: logistic anyitn i.irs
	

	** GENERATE ANY NET and NET ACCESS FOR PLACES LIKE MYANMAR
	** in HR file
	clonevar numnet=hml1
	gen anynet=numnet>0
	
	keep hhid numnet anynet
	save numnet, replace
	
	** open PR file
	merge m:1 hhid using numnet
	drop _merge
	
	gen potusenet=numnet*2
	label var potusenet "potential net users in hh"


	bysort hhid: gen accessnet = potusenet/defacto
	replace accessnet=1 if potusenet/defacto >1
	label var accessnet "proportion in hh with access to net of any kind"
	
	
	
	


